La Columna de Matemática Computacional

Sección a cargo de

Tomás Recio

El objetivo de esta columna es presentar de manera sucinta, en cada uno de los números de La Gaceta*, alguna cuestión matemática en la que los cálculos, en un sentido muy amplio, tengan un papel destacado. Para cumplir este objetivo el editor de la columna (sin otros méritos que su interés y sin otros recursos que su mejor voluntad) quisiera contar con la colaboración de los lectores, a los que anima a remitirle (a la dirección que se indica al pie de página[1]) los trabajos y sugerencias que consideren oportunos.*

En este número...

...presentamos un artículo de los profesores García Capitán y Romero Márquez, que incluye el planteamiento y resolución de diversos y complejos problemas de geometría elemental usando coordenadas baricéntricas y programas de cálculo simbólico.

Los profesores García Capitán y Romero Márquez, trabajan y –en muchos casos– colaboran desde hace tiempo en la resolución de diversos problemas, propuestos en las secciones correspondientes de La Gaceta, *American Mathematical Monthly*, *Crux Mathematicorum*, *Laboratorio Virtual de Triángulos con Cabri*, *Revista Escolar de la Olimpiada Iberoamericana de Matemática*, etc. La lista de publicaciones, en papel o electrónicas, que dedican secciones a problemas de enunciado elemental –en un sentido de la palabra– pero de dificultad manifiesta, es muy extensa y prueba la vitalidad y persistencia de la especie matemática de los *problemistas*, de la que los dos autores de esta *Columna de Matemática Computacional* son dos excelentes ejemplos.

El desarrollo de programas de cálculo literal con ordenador y de geometría dinámica ha supuesto una revolución en este ámbito de la resolución de problemas elementales, facilitando la realización de los cálculos y dibujos, favoreciendo la experimentación y permitiendo al *problemista* concentrarse en los aspectos más creativos y difíciles del problema que tiene entre sus manos, como han hecho nuestros autores.

En otro sentido, esos mismos recursos informáticos fomentan el desarrollo, aún incipiente, de técnicas de demostración automática de problemas similares a los que aquí se contemplan. El libro de geometría plana de Doron Zeiberger (¡escrito en 2050!) que puede descargarse en `www.math.rutgers.edu/ zeilberg/GT.html` es un ejemplo interesante, que intentaremos traer a esta columna en otra ocasión.

[1]Tomás Recio. Departamento de Matemáticas. Facultad de Ciencias. Universidad de Cantabria. 39071 Santander. `tomas.recio@unican.es`

Planteando y resolviendo problemas de Geometría con *Mathematica*

por

Juan Bosco Romero Márquez y Francisco Javier García Capitán

1 . Introducción

En estas páginas volvemos sobre los enunciados de algunos problemas y ahondamos en ellos, usando como herramientas las coordenadas baricéntricas y el programa de cálculo simbólico *Mathematica*.

Consideramos un triángulo de referencia ABC, cuyos lados son $a = BC$, $b = CA$, $c = AB$. Las coordenadas baricéntricas de un punto cualquiera P son tres números x, y, z proporcionales a las áreas (con signo) de los triángulos PBC, PCA, PAB respectivamente.

Por ejemplo, las coordenadas baricéntricas del baricentro G son $(1 : 1 : 1)$, y las del incentro I son $(a : b : c)$. Compárese su sencillez a las correspondientes coordenadas cartesianas en términos de los vértices A, B y C.

Creemos que la obra de de referencia para cualquier lector interesado en las coordenadas baricéntricas para aplicarlas a investigaciones sobre el triángulo es Yiu [10]. Si sólo se busca una primera introducción, puede ser suficiente la lectura de García Capitán [1, 2].

Aunque en la mayoría de las expresiones que obtenemos son polinómicas, los cálculos pueden ser tediosos y repetitivos, por lo que puede ser conveniente usar un programa de cálculo simbólico, como *Mathematica*. De esta manera el cálculo de puntos, rectas, circunferencias, cónicas, comprobación de alineaciones y concurrencias, etc. puede automatizarse.

Para usar las coordenadas baricéntricas con *Mathematica* tendremos presentes las rutinas definidas en el cuaderno *Baricentricas.nb*, que puede descargarse de `http://garciacapitan.auna.com/baricentricas`.

De la implementación con *Mathematica* digamos aquí solamente que

1. Tanto un punto como una recta serán para nosotros una terna `{p,q,r}`, que representará igualmente al punto $(p : q : r)$ como a la recta $px + qy + rz = 0$.

2. La recta que pasa por dos puntos y el punto de intersección de dos rectas se calcula de la misma forma, hallando el producto vectorial de las ternas correspondientes (usando la función `Cross`).

3. La comprobación de alineación de tres puntos o de concurrencia de tres rectas se reduce a comprobar que se anula el determinante formado por las ternas correspondientes (usando la función `Det`).

2 . Problemas propuestos

2.1 . Un resultado de alineación y su recíproco

Consideremos el siguiente problema, que puede encontrarse como ejercicio en Lachlan ([6], pág 59):

Problema 1 *Sean ABC un triángulo y P un punto del plano. Trazamos las rectas perpendiculares en P a AP, BP y CP, respectivamente, y estas rectas cortan a las rectas BC, CA y AB en U, V, W, también respectivamente. Entonces U, V y W están alineados.*



Figura 1: U, V y W están alineados.

La comprobación del enunciado es fácil con ayuda de *Mathematica*:

```
<<Baricentricas‘; ptP={x,y,z};
ptU=Punto[Recta[ptB,ptC],Perpendicular[ptP,ptP,ptA]];
ptV=Punto[Recta[ptC,ptA],Perpendicular[ptP,ptP,ptB]];
ptW=Punto[Recta[ptA,ptB],Perpendicular[ptP,ptP,ptC]];
Det[{ptU,ptV,ptW}]

0
```

Pretendemos ahora generalizar este problema: ¿Siguen estando alineados U, V y W si las rectas UP, VP y WP forman con PA, PB, PC un mismo ángulo arbitrario α? (en el Problema 1 tenemos $\alpha = 90°$).

Para ello usamos la función `CuartaRecta[P,r1,r2,r3]`, que implementa una fórmula de Paul Yiu para hallar la recta `r4` que pasa por el punto `P` y que forma con la recta `r3` el mismo ángulo que `r1` forma con `r2`. Para ver detalles sobre esta fórmula de Paul Yiu puede consultarse [3].

```
ptP = {x, y, z}; ptU = {0, q, r}; ptV = Punto[CuartaRecta[ptP,
Recta[ptP, ptU],
              Recta[ptP, ptA], Recta[ptP, ptB]], Recta[ptC, ptA]];
ptW = Punto[CuartaRecta[ptP, Recta[ptP, ptU],
              Recta[ptP, ptA], Recta[ptP, ptC]], Recta[ptA, ptB]];

Factor[Det[{ptU, ptV, ptW}]]
```

El resultado es de la forma

$$qrx^2\left(-yza^2+c^2y^2+b^2z^2+b^2yz+c^2yz\right)$$
$$\left(rS_By^2+qS_Cz^2-(c^2q+rS_A)xy-(b^2r+qS_A)xz-(rS_C+qS_B)yz\right).$$

El primer paréntesis puede comprobarse que nunca se anula y el segundo corresponde a la circunferencia con diámetro AU. Esto indica que para que los puntos U, V, W estén alineados es necesario que el ángulo APU sea recto, por lo que podríamos resumir

PROBLEMA 1* *Sean ABC un triángulo y P un punto del plano. Trazamos rectas que forman en P un mismo ángulo α con AP, BP y CP, respectivamente, y estas rectas cortan a las rectas BC, CA y AB en U, V, W, también respectivamente. Entonces U, V y W están alineados si y solo si $\alpha = 90°$.*

2.2 . ¿DEBE SER RECTÁNGULO?

Consideremos ahora este problema, propuesto por Romero Márquez [8] en *Crux Mathematicorum*:

PROBLEMA 2 *Sea ABC un triángulo no isósceles, con $AC > AB$ y rectángulo en A. Sea D el pie de la altura trazada desde A hasta BC. Sea G el punto de intersección de la recta AD con la paralela por C a AB. Sea E el punto tal que $ACGE$ es un rectángulo, y sea F el punto tal que $BFGE$ es un rectángulo. Sea H la intersección de AG y BF. Sean X la intersección de las diagonales del cuadrilátero $CDHF$ e Y la intersección de las diagonales del cuadrilátero $BDGE$. Demostrar que los triángulos ABC, DFE y DXY son semejantes.*



Figura 2: El triángulo ABC es rectángulo y AD es la altura.

Podemos intentar generalizar el problema en varias direcciones: ¿Es obligatorio que sea el triángulo rectángulo en A? ¿Es obligatorio que la ceviana AD sea una altura?

Suprimiendo la condición de triángulo rectángulo y la de que AD sea la altura nos planteamos la siguiente

Conjetura *Sean ABC un triángulo cualquiera y AD una ceviana trazada desde A sobre el lado BC. Sea G el punto de intersección de la recta AD (prolongada) hasta que corte a la recta que pasa por C y es paralela AB. Sea el punto E sobre AB (prolongado) tal que $ACGE$ es un paralelogramo, y sea F el punto sobre CG tal que $BFGE$ es un paralelogramo. Sea H la intersección de AG y BF. Sea X la intersección de las diagonales del cuadrilátero $CDHF$, y sea Y la intersección de las diagonales del cuadrilátero $BDGE$. ¿Serán DXY y DFE semejantes a ABC en algún caso además del establecido por el Problema 2?*

Si hacemos

```
ptD = {0, v, w};
ptG = Punto[Recta[ptA,ptD],Paralela[ptC,ptA,ptB]];
ptE = Punto[Recta[ptA, ptB], Paralela[ptG, ptA, ptC]];
ptF = Punto[Recta[ptC, ptG], Paralela[ptB, ptA, ptC]];
ptH = Punto[Recta[ptA, ptG], Recta[ptB, ptF]];
ptX = Punto[Recta[ptD,ptE], Recta[ptB, ptG]];
ptY = Punto[Recta[ptD,ptF], Recta[ptH, ptC]];
Punto[Recta[ptX, ptY],Recta[ptE, ptF]]
```

obtenemos que el punto $Z = XY \cap FE$ tiene coordenadas $(v-2w : -v+w : w)$, con suma cero, por lo que Z es infinito, es decir XY y FE son rectas paralelas y los triángulos DXY y DFE son semejantes para cualquier triángulo ABC y cualquier ceviana AD.

Ahora queremos que el triángulo DEF sea también semejante a ABC. Para ello hacemos

```
Factor[c^2 CuadradoDistancia[ptD,ptE]-b^2
CuadradoDistancia[ptD,ptF]];
Factor[a^2 CuadradoDistancia[ptE,ptD]-b^2
CuadradoDistancia[ptE,ptF]]
```

Cuando el factor $\Delta = b^2vw - c^2v^2 + (b^2 + c^2 - a^2)w^2$ sea nulo, las dos expresiones anteriores se anularán y los triángulos ABC y DEF serán semejantes.

Si queremos que esto ocurra cuando AD es la altura, tenemos en cuenta que el pie de la altura trazada por A tiene coordenadas $(0 : S_C : S_B)$ y hacemos $v = S_C, w = S_B$, resultando

$$\Delta = \frac{a^2}{4}\left(-a^2 + b^2 + c^2\right)\left(a^2 - 3b^2 + 3c^2\right),$$

es decir, la propiedad se cumple siendo AD la altura no sólo para los triángulos rectángulos en A sino para los que cumplen la condición $3b^2 = a^2 + 3c^2$.

Para visualizar estos triángulos fijamos el lado BC de manera que $B = (0, 0)$ y $C = (1, 0)$ (y por tanto $a = 1$). El vértice A lo consideramos variable, con coordenadas (x, y), por lo que tendremos

$$b = \sqrt{(x-1)^2 + y^2}, \quad c = \sqrt{x^2 + y^2},$$

y así resulta que $a^2 - 3b^2 + 3c^2 = 2(3x - 1)$.

Igualando a cero esta expresión obtenemos $x = \frac{1}{3}$, de donde deducimos que una condición suficiente para que la ceviana AD cumpliendo el enunciado sea la altura es suficiente que D divida al segmento BC en la razón $1 : 2$.

En general si imponemos que AD es una determinada ceviana, aparecerá un lugar geométrico para A. En el caso anterior, si fijamos los vértices B y C, el lugar geométrico para el punto A está formado por la circunferencia con diámetro BC (que corresponde al triángulo rectángulo) y la recta perpendicular a BC trazada por el punto D sobre BC tal que $BD : DC = 1 : 2$.



Figura 3: AD es la altura, pero ABC no es rectángulo.

2.3 . Analizando una desigualdad

Veamos en este apartado cómo analizar una desigualdad que aparece en otro problema publicado en Crux Mathematicorum, en este caso propuesto por Toshio Seimiya [9]:

Problema 3 *Sea ABC un triángulo rectángulo con el ángulo recto en A. Sea D el pie de la perpendicular desde A a BC, y sean E y F las intersecciones de la bisectriz del $\angle B$ con AD y AC respectivamente. Demostrar que $DC > 2EF$.*

Queremos analizar si la desigualdad $DC > 2EF$ se cumple o no para otros triángulos ABC aparte de los rectángulos en A. Hacemos:

```
ptD = Pie[ptA, ptB, ptC];
ptE = Punto[Recta[ptB, ptI], Recta[ptA, ptD]];
ptF = Punto[Recta[ptB, ptI], Recta[ptA, ptC]];
Factor[CuadradoDistancia[ptC, ptD] - 4
CuadradoDistancia[ptE, ptF]]
```

El resultado es

$$\frac{\left(a^2+b^2-c^2\right)^2\left(a^4+4ca^3-b^2a^2+6c^2a^2-12c^3a-2b^2ca+c^4-b^2c^2\right)}{4a^2(a+c)^2(a-b+c)(a+b+c)}$$

El signo de esta expresión depende del signo que tenga el paréntesis grande del numerador. Si, como antes, fijamos los vértices $B=(0,0)$ y $C=(1,0)$

podemos obtener el lugar geométrico del punto $A = (x, y)$ de manera que el triángulo ABC cumpla el enunciado.

Al hacer en el mencionado paréntesis la sustitución

$$a \to 1, b \to \sqrt{(x-1)^2 + y^2}, c \to \sqrt{x^2 + y^2}$$

obtenemos la ecuación

$$x^3 + 2x^2 + y^2 x + x + 2y^2 = \sqrt{x^2 + y^2}\left(7x^2 + 7y^2 - 2x - 1\right).$$

La figura 4 muestra la gráfica que da `ImplicitPlot` de la función obtenida. Además hemos representado la circunferencia con diámetro BC, correspondiente a los casos en que ABC es rectángulo en A, y la perpendicular al segmento BC por C, correspondiente a los casos en que el triángulo ABC es rectángulo en ABC. La región sombreada corresponde a los puntos en los que se cumple la desigualdad del enunciado.



Figura 4. Los puntos A para los que $DC > 2 \cdot EF$.

2.4 . Sorpresa de isósceles

Veamos otro problema de Romero Márquez, publicado con el número 2797 en Crux Mathematicorum [7]. Sobre este problema también puede consultarse Honsberger [5].

PROBLEMA 3 *En el triángulo ABC, supongamos que AD es una altura. Supongamos que las perpendiculares desde D cortan a los lados AB y AC en en E y F, respectivamente. Supongamos que G y H son puntos sobre AB y AC tales que $DG \parallel AC$ y $DH \parallel AB$. Demostrar que* (a) *EF y GH se cortan en un punto A^* sobre BC.* (b) *Definiendo B^* y C^* de forma similar, demostrar que A^*, B^* y C^* están alineados.*



Figura 5: A^* está sobre BC.

En la solución de este problema, el editor de Crux ofrece una generalización para cualquier ceviana AD, no necesariamente una altura. Nosotros ofrecemos aquí esta otra versión:

PROBLEMA 4 *Sea ABC un triángulo. Sean P un punto que no está sobre los lados y D el punto en que la ceviana AP corta a la recta BC. Supongamos que las paralelas por D a AB y AC cortan a las rectas AC y AB en H y G, respectivamente. A continuación, sean las intersecciones: $I = GH \cap AD, J = BI \cap GD, K = CI \cap DH$, y E, F intersecciones con AB y AC de las paralelas a CI y BI respectivamente. Entonces:*

1. *Los puntos E, J, K, F están alineados y la recta que los contiene corta a la recta GH en un punto A^* sobre BC.*

2. *Si B^*, C^* se construyen de forma similar, los puntos A^*, B^* y C^* están alineados.*



Figura 6: ¿Cuando es AMN isósceles?

Introducimos los datos del problema de la siguiente manera:

```
ptP = {u, v, w};
ptD = Punto[Recta[ptA, ptP], rtBC];
ptG = Punto[Paralela[ptD, rtCA], rtAB];
ptH = Punto[Paralela[ptD, rtAB], rtCA];
ptI = Punto[Recta[ptG, ptH], Recta[ptA, ptD]];
ptE = Punto[Paralela[ptD, ptI, ptC], rtAB];
ptF = Punto[Paralela[ptD,ptI, ptB], rtCA];
ptJ = Punto[Recta[ptI, ptB], Recta[ptD, ptG]];
ptK = Punto[Recta[ptI, ptC], Recta[ptD, ptH]];
EstanAlineados[{ptE, ptJ, ptK, ptF}]
ptAe = Punto[Recta[ptJ, ptK], Recta[ptG, ptH]]
```

La comprobación de que A^* (representado por `ptAe`) está sobre BC la hacemos comprobando que tiene primera coordenada 0.

Así obtenemos *True* para la comprobación de alineación y que $A^* = (0 : -v^2 : w^2)$. También vemos en la figura que las rectas GD, IB, EF, así como las rectas HD, IC y EF, con concurrentes. Por ejemplo, hacemos:

```
Det[{Recta[ptG, ptD], Recta[ptB, ptI], Recta[ptE, ptF]}]
0
```

Llamando J, K a los puntos de concurrencia de las rectas anteriores, hallamos la intersección M, N de AJ, AK con BC.

Nos preguntamos, ¿cuándo será isósceles el triángulo AMN?

De nuevo cálculos con *Mathematica* indican que el triángulo AMN será isósceles sólo si la ceviana AD es una altura.

```
ptM = Punto[Recta[ptA, ptJ], rtBC];
ptN = Punto[Recta[ptA, ptK], rtBC];
Factor[CuadradoDistancia[ptA, ptM] -
CuadradoDistancia[ptA, ptN]]
```

El resultado es la expresión

$$\frac{2vw\left(va^2 - wa^2 - b^2v + c^2v - b^2w + c^2w\right)}{(v+w)^3},$$

y el numerador se anulará cuando $(0 : v : w) = (0 : S_C : S_B)$, es decir, cuando AD sea la altura.

2.5 . Uno nuevo

Consideremos como punto de partida de nuestra investigación un triángulo ABC y un punto P. Llamemos a H_a, H_b, H_c a los ortocentros de los triángulos PBC, PCA, PAB. ¿Qué propiedades tiene el triángulo $H_aH_bH_c$?

Escribimos

```
ptP = {u, v, w};
ptHa := Ortocentro[ptP, ptB, ptC];
ptHb := Ortocentro[ptP, ptC, ptA];
ptHc := Ortocentro[ptP, ptA, ptB];
PerspectivoConABC[{ptHa, ptHb, ptHc}]
```

Obtenemos

```
-2 (-a - b + c) (a - b + c) (-a + b + c) (a + b + c)
 (-a^2 u + b^2 u + c^2 u - a^2 v + b^2 v - c^2 v) (u + v + w)
 (-a^2 u + b^2 u + c^2 u - a^2 w - b^2 w + c^2 w) (a^2 v - b^2 v + c^2 v - a^2 w - b^2 w + c^2 w)
 (c^2 u v + b^2 u w + a^2 v w) == 0
```

Las expresiones obtenidas corresponden a la recta del infinito, a las tres alturas del triángulo ABC y a la circunferencia circunscrita al triángulo ABC.

Para calcular el área del triángulo $H_aH_bH_c$ hacemos

```
Simplify[AreaTriangulo[ptHa, ptHb, ptHc]]

1
```

comprobando que el triángulo $H_aH_bH_c$ tiene el mismo área que el triángulo ABC.

Supongamos ahora que P está sobre la circunferencia circunscrita a ABC. Estudiemos quién es el centro de perspectiva Q de los triángulos $H_aH_bH_c$ y

ABC. Para considerar un punto arbitrario P sobre la circunferencia circunscrita usando coordenadas baricéntricas tenemos en cuenta que esta circunferencia es la conjugada isogonal de la recta del infinito (ver por ejemplo el apartado 5.2 de Yiu [10]).

Entonces hacemos

```
ptP = ConjugadoIsogonal[{-v - w, v, w}]
ptQ = Factor[PerspectorConABC[{ptHa, ptHb, ptHc}]]
```

$$\{a^2\, v\, w,\ b^2\ (-v-w)\ w,\ c^2\ v\ (-v-w)\}$$

$$\{(a^2\, v - b^2\, v + c^2\, v - a^2\, w - b^2\, w + c^2\, w)\\ (-a^2\, c^2\, v - b^2\, c^2\, v + c^4\, v + a^2\, b^2\, w - b^4\, w + b^2\, c^2\, w),\\ -(a^2\, v - b^2\, v - c^2\, v - 2\, b^2\, w)\\ (-a^2\, c^2\, v - b^2\, c^2\, v + c^4\, v + a^4\, w - a^2\, b^2\, w - 2\, a^2\, c^2\, w - b^2\, c^2\, w + c^4\, w),\\ -(-2\, c^2\, v + a^2\, w - b^2\, w - c^2\, w)\\ (a^4\, v - 2\, a^2\, b^2\, v + b^4\, v - a^2\, c^2\, v - b^2\, c^2\, v - a^2\, b^2\, w + b^4\, w - b^2\, c^2\, w)\}$$

Aunque las coordenadas del punto Q sean complicadas podemos usar la función `Eliminate` de *Mathematica* para hallar el lugar geométrico de Q cuando P varía sobre la circunferencia:

```
lugar = Eliminate[{x, y, z} == ptQ, {v, w}]
```

$$(a^2 - b^2 - c^2)\ x^2 + x\ (2\, c^2\, y + 2\, b^2\, z) ==\\ a^2\, y^2 - b^2\, y^2 + c^2\, y^2 - 2\, a^2\, y\, z + a^2\, z^2 + b^2\, z^2 - c^2\, z^2$$

Vemos que es una cónica. Si ahora escribimos

```
Solve[lugar /. x → 0, z]
```

$$\left\{\{z \to y\},\ \left\{z \to \frac{a^2\, y - b^2\, y + c^2\, y}{a^2 + b^2 - c^2}\right\}\right\}$$

vemos que los puntos de intersección de con la recta BC son el punto medio de BC y el pie de la altura trazada por A. En consecuencia, deducimos que el lugar geométrico del punto Q es la circunferencia de los nueve puntos del triángulo ABC.

Todo esto nos lleva a plantear el siguiente

Problema 5 *Dados un triángulo ABC y un punto P, consideramos los triángulos PBC, PCA, PAB y sus ortocentros H_a, H_b, H_c. Entonces:*

1. *Los puntos H_a, H_b y H_c nunca están alineados, a menos que P esté sobre uno de los lados del triángulo ABC.*

2. *Las rectas AH_a, BH_b y CH_c son concurrentes si y solo si P está sobre alguna de las alturas del triángulo ABC o en la circunferencia circunscrita al triángulo ABC.*

3. *El triángulo $H_aH_bH_c$ siempre tiene la misma área que ABC.*

4. *Cuando P está sobre la circunferencia circunscrita a ABC, el punto de concurrencia Q de AH_a, BH_b y CH_c está sobre la circunferencia de los nueve puntos del triángulo ABC y el triángulo $H_aH_bH_c$ es el resultado de aplicar a ABC una simetría central de centro Q. De hecho, en este caso Q es el punto medio del punto dado P y el ortocentro H del triángulo ABC.*



Figura 7: El lugar de Q es la circunferencia de los nueve puntos del triángulo ABC.

El problema anterior y el que aparece en la siguiente sección, una vez que han sido obtenidos mediante estos cálculos con *Mathematica*, quedan a disposición de los lectores, si es posible para resolverlos de forma sintética.

2.6 . Perpendiculares sobre el triángulo mediano

Para finalizar esta serie de problemas consideremos un triángulo ABC y su triángulo mediano $A'B'C'$. Por los puntos A', B', C' trazamos perpendiculares a las medianas AA', BB', CC' respectivamente, y estas perpendiculares determinan un triángulo XYZ. ¿Qué propiedades podemos ver en este triángulo XYZ?



Figura 8: Perpendiculares sobre el triángulo mediano.

Por ejemplo, para averiguar en qué casos el triángulo XYZ resultará rectángulo en X hacemos

```
{ptA1, ptB1, ptC1} = TrianguloCeviano[ptG];
ptX = Punto[Perpendicular[ptB1, ptB, ptB1],
   Perpendicular[ptC1, ptC, ptC1]];
ptY = PermutarTerna[ptX];
ptZ = PermutarTerna[ptY];
dYZ = CuadradoDistancia[ptY, ptZ];
dZX = CuadradoDistancia[ptZ, ptX];
dXY = CuadradoDistancia[ptX, ptY];
Factor[dXY + dZX - dYZ]
```

$$-\frac{(5 a^2 - b^2 - c^2)\ (a^2 + b^2 + c^2)^2}{9\ (a - b - c)\ (a + b - c)\ (a - b + c)\ (a + b + c)}$$

Vemos que la condición es $5a^2 = b^2 + c^2$. Si consideramos fijos los vértices $B = (-1, 0)$ y $C = (1, 0)$ de estos triángulos y dejamos que $A = (x, y)$ varíe sometido a esta condición, obtendremos el lugar geométrico de A:

```
Factor[5 a^2 - b^2 - c^2 /.
   {a → 2, b → Sqrt[(x - 1)^2 + y^2], c → Sqrt[(x + 1)^2 + y^2]}]
```

$$-2\,(-9 + x^2 + y^2)$$

Resulta entonces que A estará sobre una circunferencia con centro el origen y radio 3. Esto nos lleva al planteamiento del siguiente problema:



Figura 9: XYZ es rectángulo en X.

Problema 6 *Dados dos puntos B y C consideramos el simétrico D de C respecto de B y el simétrico E de B respecto de C, y un punto cualquiera A sobre la circunferencia de diámetro DE. Entonces:*

1. *Las medianas BB' y CC' son perpendiculares.*
2. *Las perpendiculares por B' y C' a las medianas BB' y CC' se cortan en un punto X sobre la mediana AA' de manera que $AX : XA' = 1 : 2$.*
3. *Si las perpendiculares por B' y A' a las medianas BB' y AA' se cortan en Z y las perpendiculares a C' y A' a las medianas CC' y AA' se cortan en Y, entonces el triángulo XYZ es rectángulo.*
4. *Las rectas XY y XZ pasan siempre por los puntos medios M y N de los segmentos DB y CE, respectivamente.*

Referencias

[1] F.J. García Capitán, *Coordenadas Baricéntricas*, disponible en pdf en `http://garciacapitan.auna.com/baricentricas`

[2] F.J. García Capitán, *Problema 245 de* triánguloscabri, disponible en `www.aloj.us.es/rbarroso/trianguloscabri/sol/sol245garcap/sol245garcap.htm`

[3] F.J. García Capitán, *Problema 309 de* triánguloscabri, disponible en `www.aloj.us.es/rbarroso/trianguloscabri/sol/sol309garcap/sol309garcap.htm`

[4] M. de Guzman Ozámiz, *La experiencia de descubrir en Geometría.* Nivola, libros y ediciones. Madrid, 2002.

[5] R. Honsberger, *Mathematical Diamonds* (Dolciani Mathematical Expositions). The Mathematical Association of America (January 2, 2003).

[6] R. Lachlan, *An Elementary Treatise on Modern Pure Geometry.* Macmillan, 1893. Libro disponible en la Universidad de Cornell: `http://historical.library.cornell.edu/math/`

[7] J.B. Romero Márquez, Problema 2797, *Crux Mathematicorum* **29** (2003) 526–528.

[8] J.B. Romero Márquez, Problema 2973, *Crux Mathematicorum* **31** (2005) 465–466.

[9] T. Seimiya, Problema 1601, *Crux Mathematicorum* **17** (1991) 13.

[10] P. Yiu, *An Introduction to the Geometry of the Triangle, 2001*, disponible en postscript en `www.math.fau.edu/yiu/geometry.html`

Juan-Bosco Romero Márquez
Colaborador honorífico adscrito al Departamento de Álgebra, Geometría y Topología
Universidad de Valladolid
Correo electrónico: `jbromeroma@gmail.com`

Francisco Javier García Capitán
Departamento de Matemáticas
I.E.S. Álvarez Cubero, Priego de Córdoba
Correo electrónico: `garciacapitan@auna.com`